# $Id: PKGBUILD 179888 2013-03-10 21:31:07Z andrea $
# Maintainer: Andrea Scarpino <andrea@archlinux.org>

pkgname=kwebkitpart
pkgver=1.3.2
pkgrel=2
pkgdesc="A WebKit browser component for KDE"
url="https://projects.kde.org/projects/extragear/base/kwebkitpart/"
arch=('i686' 'x86_64')
license=('LGPL')
depends=('kdelibs')
makedepends=('cmake' 'automoc4')
install=${pkgname}.install
source=("ftp://ftp.archlinux.org/other/packages/${pkgname}/${pkgname}-${pkgver}.tar.bz2")
md5sums=('2df1c70371b99e5f638fff702f789ba1')

build() {
  sed -i '/add_subdirectory(kdelauncher)/d' ${pkgname}-${pkgver}/CMakeLists.txt

  mkdir build
  cd build
  cmake ../${pkgname}-${pkgver} \
    -DCMAKE_INSTALL_PREFIX=/usr \
    -DCMAKE_BUILD_TYPE=Release
  make
}

package() {
  cd build
  make DESTDIR="${pkgdir}" install
}
